Non-adaptive model asymptotic competitive ratio

This file complements the paper “Scheduling with a processing time oracle” by Dufossé, Dürr, Nadal, Trystram and Vásquez.

Algorithm chooses r>0 and 0 <= α < 1.
Adversary chooses 0 <= β <= α and 0 <= γ < 1-α.

The upper bounds are strict inequalities because the last touched job is always executed and short.

We have n jobs of length either p or p+x.
Algorithm decides to test the first α fraction of the jobs and to execute untested the remaining *1-*α fraction.
Adversary decides to return a fraction β long jobs to the tests and γ fraction of long jobs to the executions.

Fractions of jobs

```
ALG
| α = #T                  | 1-α = #E                     |
| β = #Tx | α-β = #Tp | γ    = #Ex | 1-α-γ = #Ep | β = #delayed x |

OPT
| 1-β-γ = #Ep             | β+γ = #Ex                    |
```

First we give the expressions of the costs of the schedules.
We only keep the n^2 dependent part, and multiply by 2 for simplicity.
Then we set the goal expression G, which is >=0 iff the ratio is at most 1+r.

In[39]:=
```
ClearAll[ALG, OPT, G]
ALG = FullSimplify[(2 β
      + (1 + p) (α - β) ^2
      + 2 (1 + p) (α - β) (1 - α + β)
      + (p + x) γ^2
      + 2 (p + x) γ (1 - α + β - γ)
      + p (1 - α - γ) ^2
      + 2 p (1 - α - γ) β
      + (p + x) β^2)];
OPT = FullSimplify[(p (1 - β - γ) ^2
      + 2 p (1 - β - γ) (β + γ)
      + (p + x) (β + γ) ^2)];
G = FullSimplify[(1 + r) OPT - ALG]
```

Out[42]= $p\,r + \alpha^2 + \beta^2 - 2\,\alpha\,(1 + \beta) + 2\,x\,\gamma\,(-1 + \alpha + \gamma) + r\,x\,(\beta + \gamma)^2$

All right. Now the algorithm fixes r and α. In response the adversary tries to minimize G by choosing appropriate β and γ. We use second order analysis for this. And break it into two cases.

Case x ≥ 2+1/p

Optimize β

β^2 has coefficient (1+rx) >0 in G, hence the extreme point of G in β is a minimum.

In[43]:=
```
bsol = Solve[D[G, β] == 0, β]
```

Out[43]= $\left\{\left\{\beta \to \frac{\alpha - r\,x\,\gamma}{1 + r\,x}\right\}\right\}$

Set β to the extreme value. Our new goal is G2.

In[44]:=
```
G2 = Expand[G /. bsol[[1]]]
```

Out[44]= $p\,r - 2\,\alpha + \alpha^2 + \frac{\alpha^2}{(1 + r\,x)^2} + \frac{r\,x\,\alpha^2}{(1 + r\,x)^2} - \frac{2\,\alpha^2}{1 + r\,x} - 2\,x\,\gamma + 2\,x\,\alpha\,\gamma - \frac{2\,r\,x\,\alpha\,\gamma}{(1 + r\,x)^2} - \frac{2\,r^2\,x^2\,\alpha\,\gamma}{(1 + r\,x)^2} + \frac{4\,r\,x\,\alpha\,\gamma}{1 + r\,x} + 2\,x\,\gamma^2 + r\,x\,\gamma^2 + \frac{r^2\,x^2\,\gamma^2}{(1 + r\,x)^2} + \frac{r^3\,x^3\,\gamma^2}{(1 + r\,x)^2} - \frac{2\,r^2\,x^2\,\gamma^2}{1 + r\,x}$

Optimize γ

G2 is quadratic in γ. What is the coefficient of γ^2 in G2?

In[45]:=
```
FullSimplify[D[D[G2, γ], γ]]
```
Out[45]= $2\,x\left(2+\frac{r}{1+r\,x}\right)$

Good, G2 is convex in γ. Hence it is minimized at the extreme value of γ.

In[46]:=
```
gsol = FullSimplify[Solve[D[G2, γ] == 0, γ]]
```
Out[46]= $\left\{\left\{\gamma\to\frac{1+r\,x-(1+r+r\,x)\,\alpha}{2+r+2\,r\,x}\right\}\right\}$

Our new goal is G3.

In[47]:=
```
G3 = Simplify[G2 /. gsol[[1]]]
```
Out[47]= $-\frac{-p\,r\,(2+r+2\,r\,x)+r\,x^2\,(-1+\alpha)^2+2\,(2+r)\,\alpha+x\left(1+2\,(-1+r)\,\alpha+\alpha^2\right)}{2+r+2\,r\,x}$

Optimize α

The goal is quadratic in α. What is the coefficient of α^2 in G3?

In[48]:=
```
D[D[G3, α], α]
```
Out[48]= $-\frac{2\,x+2\,r\,x^2}{2+r+2\,r\,x}$

Good, hence the algorithm chooses extreme value for α.

In[49]:=
```
asol = Simplify[Solve[D[G3, α] == 0, α]]
```
Out[49]= $\left\{\left\{\alpha\to\frac{-2+x+r\left(-1-x+x^2\right)}{x\,(1+r\,x)}\right\}\right\}$

In[50]:=
```
G4 = Simplify[G3 /. asol[[1]]]
```
Out[50]= $\frac{2+r-2\,x+p\,r\,x-r\,x^2+p\,r^2\,x^2}{x+r\,x^2}$

Optimize r

In[51]:=
```
FullSimplify[D[D[G4, r], r]]
```
Out[51]= $-\frac{2\,(-1+x)^2}{(1+r\,x)^3}$

As expected, the goal is concave in r. So probably the goal has two roots.

In[52]:=
```
rsol = Simplify[Solve[G4 == 0, r]]
```

Out[52]=
$$\left\{\left\{r \to -\frac{1 + p\,x - x^2 + \sqrt{8\,p\,(-1 + x)\,x^2 + \left(1 + p\,x - x^2\right)^2}}{2\,p\,x^2}\right\}, \left\{r \to \frac{-1 - p\,x + x^2 + \sqrt{8\,p\,(-1 + x)\,x^2 + \left(1 + p\,x - x^2\right)^2}}{2\,p\,x^2}\right\}\right\}$$

These roots are real, since we can safely assume x>=1. First root is the smaller one.
We denote

In[53]:=
```
Δ = Numerator[r /. rsol[[1]]][[4]][[2]][[1]]
```

Out[53]=
$$8\,p\,(-1 + x)\,x^2 + \left(1 + p\,x - x^2\right)^2$$

We have Sqrt[Δ]>=x^2-px-1. Hence the first root is negative. As a consequence the algorithm chooses for r any value between 0 and the second root. Say the second one.

In[54]:=
```
rstar = r /. rsol[[2]]
```

Out[54]=
$$\frac{-1 - p\,x + x^2 + \sqrt{8\,p\,(-1 + x)\,x^2 + \left(1 + p\,x - x^2\right)^2}}{2\,p\,x^2}$$

and the ratio is 1+rstar.
Now we need to verify the conditions on α,β,γ.

Domain check on α, β,γ

Let's start with the simplest conditions to verify.

In[55]:=
```
bstar = β /. bsol[[1]]
```

Out[55]=
$$\frac{\alpha - r\,x\,\gamma}{1 + r\,x}$$

We have clearly β<=α.

In[56]:=
```
gstar = γ /. gsol[[1]]
```

Out[56]=
$$\frac{1 + r\,x - (1 + r + r\,x)\,\alpha}{2 + r + 2\,r\,x}$$

In[57]:=
```
FullSimplify[(1 - α) - gstar]
```

Out[57]=
$$\frac{1 + r + r\,x - (1 + r\,x)\,\alpha}{2 + r + 2\,r\,x}$$

Ok, by α<1, this is positive. Hence γ<1-α as required.
But β or even γ could be negative. Meanwhile since we set all those values to their extreme points, the expressions are simpler.

In[58]:=
```
bstar = FullSimplify[β /. bsol[[1]] /. gsol[[1]] /. asol[[1]] /. rsol[[2]]]
```

Out[58]= $-\frac{1 + p x + x^2 - \sqrt{8 p (-1 + x) x^2 + (1 + (p - x) x)^2}}{2 x^2}$

We have b* = r* p - 1. See below.

In[59]:=
```
FullSimplify[rstar p - 1 - bstar]
```

Out[59]= 0

In other words condition β>=0 translates into r* >= 1/p. Now let's verify this later condition instead.

In[60]:=
```
Simplify[Solve[rstar == 1 / p, x]]
```

Out[60]= $\left\{\left\{x \to 2 + \frac{1}{p}\right\}\right\}$

All right, this realized at x=2+1/p and only there. What happens for larger values of x?

In[61]:=
```
Limit[rstar, x → Infinity]
```

Out[61]= $\frac{1}{p}$

Sanity check: What happens for smaller values, for example x=1?

We have simplify the expression under the condition p>0 to get something simple in Mathematica.

In[62]:=
```
Simplify[rstar /. {x → 1}, p ≥ 0]
```

Out[62]= 0

What happens when x is between 2+1/p and infinity?

Fur this purpose we analyze the slope of r* at x=2+1/p.

In[63]:=
```
Simplify[Simplify[D[rstar, x]] /. {x → 2 + 1 / p}, p ≥ 0]
```

Out[63]= $\frac{1}{6 + \frac{1}{p^2} + \frac{4}{p} + 2 p}$

All right, hence b*>=0 is satisfied for all x>=2+1/p.

Now let's verify γ>0.

In[64]:=
```
Simplify[gstar /. asol[[1]]]
```

Out[64]= $\frac{1 + r}{x + r x^2}$

ok we verified all conditions.

Case 0 ≤ x < 2+1/p

We expect that the algorithm will never test in this case. Let's see.

We computed best response β and γ as a function of α and r.

Then best choices of α and r in response.
When x is too small, the best choice of α would lead in an invalid choice β.
It actually means that the adversary cannot choose β in that case and is limited to choose β=0.
With this new choice of β, the best responses in α and r need to be recomputed.
Contrary to what we wrote currently in the paper, even when the algorithm knows that all tested jobs will be short, then it still makes sense to test some jobs.

In[76]:= `G5 = G /. {β → 0}`

Out[76]= $p\,r - 2\,\alpha + \alpha^2 + r\,x\,\gamma^2 + 2\,x\,\gamma\,(-1 + \alpha + \gamma)$

In[77]:= `g2sol = Solve[D[G5, γ] == 0, γ]`

Out[77]= $\left\{\left\{\gamma \to \frac{1-\alpha}{2+r}\right\}\right\}$

Very good choice, this is between 0 and 1-α for sure.

In[102]:= `G6 = FullSimplify[(2 + r) G5 /. g2sol[[1]]]`

Out[102]= $p\,r\,(2+r) - x\,(-1+\alpha)^2 + (2+r)\,(-2+\alpha)\,\alpha$

Is the goal function convex in α ?

In[81]:= `d2a = FullSimplify[D[D[G6, α], α]]`

Out[81]= $2\,(2 + r - x)$

Case x ≤ 2+r (small x)

In this case the algorithm chooses α=0.

In[83]:= `G7 = FullSimplify[G6 /. {α → 0}]`

Out[83]= $p\,r\,(2+r) - x$

In[91]:= `r2sol = Solve[G7 == 0, r]`

Out[91]= $\left\{\left\{r \to \frac{-p-\sqrt{p^2+p\,x}}{p}\right\}, \left\{r \to \frac{-p+\sqrt{p^2+p\,x}}{p}\right\}\right\}$

The second root is positive. And is the ratio in this case.
Let’s verify that the case condition is satisfied.
When x=0 the we have clearly x <= 2+ r = 1 + sqrt(1+x/p).
Both sides of the inequality are continuous and increasing in x. For which value of x do we have equality?

In[99]:= `Solve[x == 1 + Sqrt[1 + x / p], x]`

Out[99]= $\left\{\left\{x \to \frac{1+2\,p}{p}\right\}\right\}$

Good by case assumption this value is not reached by x. Hence condition is satisfied.

Case 2+r < x

But from the previous case assumption (one level above) we have 2+r < x < 2+1/p.

In[100]:=
```
a2sol = Solve[D[G6, α] == 0, α]
```
Out[100]=
```
{{α → 1}}
```

This means that the algorithm tests all jobs which happen to be all short. The ratio is then simply (1+p)/p=1+1/p. But then 2+r = 3+1/p which exceeds the case assumption of x < 2+1/p. So this case is void, and the analysis complete.

When 0<=x < 2+1/p then the ratio is
Sqrt[1+x/p]
and when x>=2+1/p the ratio is

$$-\frac{1+p\,x-x^2+\sqrt{8\,p\,(-1+x)\,x^2+\left(1+p\,x-x^2\right)^2}}{2\,p\,x^2}$$